PharmaSUG 2022 - Paper SA-118

Win Ratio Simulation For Power Calculation Made Easy

Developing a SAS Macro to Implement Win Ratio Approach for Study Designing Purpose

Lili Li, Chih-Chieh Chang, Benjamin Fine, and Yu-Chen Su, Inari Medical

ABSTRACT

More often than not, treatment effects cannot be evaluated by a single event. Therefore, composite endpoints have been frequently used for decision-making in clinical trials. However, conventional analysis of composite endpoints still treats each component as separate and equal. While with clinical relevance in mind, for instance, death may be seen as more important than other events such as stroke; thus, hierarchy ranking of component endpoint is desired. In 2012, Pocock et al. proposed the win ratio approach for composite endpoint analysis. In this approach, each component from the composite endpoint is ranked by its clinical importance in the study, and then analyzed by one component at a time in a hierarchical fashion between any two subjects. Since then, applications of win ratio approach have been steadily gaining momentum in clinical trial analysis. While more of those applications are observed at the hypothesis testing stage, fewer are focused on the early study design phase. This paper will present an easy-to-use SAS Macro that has the flexibility to customize component event rates and rankings based on clinical relevance to assist at the study design stage by simulating subject-level data with composite endpoints to derive the win ratio statistics and calculate power for a given sample size.

Key Words: Composite Endpoints; Win Ratio; Simulation; Power Calculation; Study Design

INTRODUCTION

Composite endpoint design has been a commonly employed approach in clinical trials especially in the cardiovascular field (Ferreira et all., 2021). The utilization of composite endpoints has been well-accepted and practiced in the process of decision-making for determining the treatment effects in a clinical trial. Conventional statistical analysis approaches for composite endpoints include using the time-to-first-event model producing a hazard ratio and 95% Confidence Interval (Cox, 1972; Andersen & Gill, 1982; Wei et al., 1989; Wei & Glidden, 1997), or a simple log-rank test in a Kaplan Meier analysis (Kaplan & Meier, 1958; Mantel, 1966). One potential shortcoming of these conventional statistical analysis approaches is that they consider all endpoint components as equal; while in real world clinical practice, that is not always the case. For instance, in conventional composite endpoint analysis, a non-fatal event carries the same clinical significance to one subject as a fatal event to another subject if both events happen at the same time-to-event point. This inability to differentiate clinical significance between composite endpoint events is, as Pocock et al. (2012) put it, “an inherent limitation” in the conventional statistical analysis methods for composite endpoints.

Pocock and his colleagues therefore proposed a new approach, now termed as the “win ratio” approach, that considers both the clinical significance and the timing of an endpoint event. Under the win ratio approach, each component of the composite endpoints is ranked by their clinical significance and relevance, and subjects are evaluated head-to-head by comparing each component on a hierarchical order under the time-to-event construct. Take some common cardiovascular composite endpoint events for example, namely, death, myocardial infarction (MI), and hospitalization, under the win ratio approach, we assign death as the highest in clinical significance when we compare in any given pair of subjects, MI the second most important, and hospitalization the least important. Using this component ranking, for the head-to-head comparison, if one subject has died while the other subject has not, then regardless of whether the other subject experiences an MI or hospitalization, the subject who dies is a “loser”. If both have reported a death, one with a shorter time-to-event is the “loser”. If neither has a death event, we

move down to compare the second significant event we ranked among the components, which in this case, the MI. Following similar logic of comparing the death event, we then decide a “winner” or a “loser” based on MI incidences, and so on so forth. If we do not find a “winner” after we exhausted comparison with all composite endpoint events ranked, we have a “tie”. With all the comparison results, we can thus derive the ratio of winners to losers for the treatment group, that is, the “win ratio (Pocock et al., 2012).

With its straightforwardness in methodology and its capacity to solve the “inherent limitation” in the conventional statistical analysis methods for composite endpoints, the win ratio approach has been gaining increasing interest and implementations in numerous clinical trial analysis (i.e., Rogers et al., 2014; Bakal et al., 2015; Abdalla et al., 2016; Dong et al., 2018; Bebu & Lachin, 2018; Yosef, Khalatbari, & Hummel, 2019; Ferreira et all., 2021). However, the existing applications of win ratio approach usually are implemented in the study outcome analysis and hypothesis testing stage, while very limited attention has been shed at the early study design stage (Peng, 2020). Moreover, there are very limited SAS applications for win ratio approach implementation, let alone win ratio simulation for power calculation at study design stage. This paper tries to provide some implementations and discussions in the gap here by presenting a SAS macro that can help biostatisticians implement simulations and power calculation using the win ratio approach for composite endpoints study at the designing stage.

METHODOLOGY

In the Pocock et al. (2012) paper, two analysis methods for the win ratio approach were discussed: the matched pair method and the unmatched pair method.

In the matched pair method, pairs of subjects are selected and determined by matching risk profile, usually a baseline risk scores and metrics. In practice, the matched pair method would require a careful and scientific evaluation of what risk profiling should be considered when looking for matching pairs. This method also always yields “unmatched” subjects where certain number of subjects would end up being excluded into the analysis, thus resulting in loss of information and data points.

On the other hand, the unmatched pair method compares one patient with every other patient in the study, both from the same arm and the different arm, without restrictions of the subjects’ risk profile. This method was originally proposed and designed by Finkelstein and Schoenfeld (1999), Pocock et al. (2012) employed this concept and extended its application to win ratio composite endpoint analysis.

For this paper, the unmatched pair method is employed. The primary reason is that, with the unmatched pair method, each subject can be compared with all other subjects in a fair comparison paradigm. In the meantime, it saves time and resources since unmatched pair method does not require correct matching variables to be identified beforehand, which is often difficult to know exactly what those variables should be in the early study designing stage. Although it is somewhat more complex to calculate the power and p values using unmatched pair method, it is more “unbiased” (Pocock et al., 2012).

This Finkelstein-Schoenfeld (F-S) test formula (Finkelstein & Schoenfeld, 1999) is followed as the foundation to construct the macro in this paper:

$$\text{F-S Score} = \frac{T}{\sqrt{V}} \qquad (1)$$

Specifically, a winner gets a score of 1 while the loser get a score of -1. If it is a tie, each subject gets a score of 0. After comparing subject *i* with other subjects in the study, we will be able to get a win-lose score U_i. A subject with higher score implies that this subject is more likely to be a winner when compared to the other subjects in the study.

$$U_i = \sum_{i \neq j} u_{ij} \qquad (2)$$

By summing all the win-lose scores for subjects in the treatment group, the T score in the F-S test ① above can be derived following the equation below (By default, $D_i = 1$ in treatment group and $D_i = 0$ in control group).

$$T = \sum_{i=1}^{N} U_i D_i \qquad ③$$

With known number of subjects in treatment group n_1 and number of subjects in control group n_2, the variance of T can be obtained as the formula below. Note, *N* is the total number of sample size in the study, with $N = n_1 + n_2$.

$$V = \frac{n_1 n_2}{N(N-1)} \sum_{i=1}^{N} U_i^2 \qquad ④$$

Once the F-S score is calculated, the power of the study per simulation can be computed by the following steps: 1) compare the F-S score to the standard normal distribution to determine the p value per each data replicate; 2) with α set to 0.05 in usual study designs, determine each data replicate as whether its p value is less or equal to 0.05; 3) derive the proportion of data replicates yielding p ≤ 0.05 based off all data replicates produced by the simulation. That proportion from step 3 is the power of the study. For instance, if 2500 data replicates are generated per the simulation and 2000 data replicates yielded p ≤ 0.05 while 500 data replicates with p above 0.05, then the power of the study is 2000/2500 = 0.80. If the goal of the desired study power is at least 80%, it can be thus determined that enough study power can be secured with the sample size per that simulation.

THE WIN RATIO SIMULATION AND POWER CALCULATION MACRO

With the methodology discussed above, the SAS macro to be presented and discussed below will provide more details to show how 1) subject-level data listing is simulated based on pre-specified component endpoint proportions, 2) the win ratio statistics are derived per simulated datasets, and 3) the power of the study is calculated based on a sufficiently large number of data replicates per the simulation.

Note, in real practice, all simulation lists need to be verified on 1) if the component events are correlated within each data replicate, and 2) if the rates of events in simulation are consistent or very close to the pre-specified component endpoint proportions per design. However, as those verifications are relatively easy, this paper will not elaborate on correlation checking or frequency checking to save space for the demonstration of key components in the macro.

KEY PARAMETERS IN THE MACRO

To ensure the macro function as designed, some key parameters need to be set up properly first. Below is an example as a quick reference:

```
%winPower (event = death mi hosp,
           ratio1 = 0.05|0.08|0.10, ratio2 = 0.09|0.12|0.20,
           subn= 100, simnum = 2500);
```

The first key parameter in the macro is the ***event*** variable. This parameter denotes what events are to be included in the designed composite endpoints. The order of the events listed in this parameter has to be consistent with the descending hierarchical order of the composite endpoints by the clinical significance. Reusing the above-mentioned common cardiovascular trials' composite endpoint events, death, MI, and hospitalization, as the example here, the configuration of ***event = death mi hosp*** means that death is the most clinically significant event, MI the second most significant event, while hospitalization is the least significant event. There is no "|" as divider in the ***event*** parameter, compared to the ratio parameters ***ratio1/ratio2***. This is specifically designed so that the parameter ***event*** can be re-employed as

intermediate data set names when merging separate simulation listings per each component to produce the complete data replicate listing. This will be illustrated more in details in the steps below.

The second set of key parameters are the ***ratio1/ratio2*** variables. The parameter ***ratio1*** corresponds to event rates in ***treatment*** arm, while ***ratio2*** corresponds to event rates in ***control*** arm. It should be emphasized here that event ratios listed in ***ratio1/ratio2*** always correspond to the pre-specified endpoint proportions, in treatment arm and control arm respectively, of the components listed in the hierarchical order as configured in the ***event*** parameter. Each proportion needs to be divided by "|" to be properly parsed and read in. Take the codes of ***ratio1 = 0.05|0.08|0.10*** above as an example, this combination means that the event rate of **death** in treatment arm is expected to be about **5%**, while the event rate of **MI** is expected to be **8%**, and the event rate of **hospitalization** is expected to be around **10%**. Similarly, ***ratio2 = 0.09|0.12|0.20*** can be interpreted that, in control arm, a reasonable guess for event rate of **death** would be **9%**, for event rate of **MI** would be **12%**, while for event rate of **hospitalization** would be **20%**.

The last set of key parameters are the ***subn*** parameter to control the sample size per arm and the parameter ***simnum*** to determine how many data replicates will be generated per each simulation run. In each data replicate, the total number of subjects will always be ***subn*2** for the consideration of a balance design between treatment and control arm. Therefore, in the codes above***, subn= 100*** and ***simnum = 2500*** mean that one complete simulation run using this configuration in this macro will produce 2500 data replicates with 100 subjects per each arm in each simulated listing.

STEP 1. PRODUCE SUBJECT-LEVEL LISTING

With the parameters pre-defined above, the subject-level listings per pre-specified component endpoint proportions can be achieved by the following codes:

```
%local i j k component proportion;
%do k = 1 %to &simnum;
%do m = 1 %to 2;
%do i = 1 %to %sysfunc(countw(&event));
%let component = %scan(&event, &i);
%let proportion = %scan(&&ratio&m, &i, '|');

    data &component._temp;
      do j = 1 to &subn;
          if j le ROUND(&subn. * &proportion.) then &component = 1;
          else &component = 0;
          x1 = rand('normal',.5,.1);
          output;
     end;
   run;

   proc sort data = &component._temp;
          by x1;
   run;

   data &component;
     length SUBJID arm $20.;
     format simn best.;
     set &component._temp;
     by x1;
     retain SUBJID_;
     if _n_ = 1 then SUBJID_ = &m. * &subn. + 1;
     else SUBJID_ = SUBJID_ +1;

     SUBJID = strip(put(SUBJID_, best.));
```

```
      simn = &k;
      if &m = 1 then arm = "treatment";
      if &m = 2 then arm = "control";

      keep SUBJID arm simn &component;
    run;

    proc sort data = &component;
          by SUBJID;
    run;
%end;
data arm_&k._&m;
  merge &event;
  by SUBJID;
run;

%end;
data simulist_&k;
  set arm_&k.:;
run;
```

Notice, to facilitate the simulation list per k^{th} data replicate, two local macro parameters are introduced (***component, proportion***) and three temporary data sets ***(&component._temp, &component, and arm_&k._&m***) are created. The local macro variable ***component*** is parsed from the ***event*** parameter configured above, to denote what endpoint event is being simulated for. While the other local macro variable ***proportion*** is translated from parameters ***ratio1/ratio2,*** depending on which treatment arm the data replicate is at, to denote what event rate is for the endpoint event being simulated for. With these two local macro variables, temporary data set ***&component._temp*** is produced to control for each event per arm with the designed proportion with a simple rand function, ***x1 = rand('normal',.5,.1)***. The step with temporary data set ***&component*** is to further add on study related accessary variables (subject identifier ***SUBJID***, and treatment names “treatment” or “control”, and data replicate identifier ***simn***) to facilitate further analysis in next steps. The temporary data set ***arm_&k._&m*** is the bridging data set for the treatment or control arm in the k^{th} data replicate. With these data sets per arm available, the combined list per k^{th} data replicate can be obtained in the last step above *shown* as ***simulist_&k.***

STEP 2. DERIVE WIN RATIO STATISTICS

The next step of the macro is to derive the win ratio statistics per each data replicate. To achieve that, an innovative yet straightforward approach is proposed: with a numeric "composite score” based on the hierarchical order of the events, this composite score will serve as the intermediate “bridge” before the final comparison to determine who is the winner in any given pair of subjects. Specifically, the pre-defined hierarchical order per the clinical significance of a composite endpoint event is translated into an exponent of 10, eventually producing a score for composite endpoint event on the scale of 1 to 10^{n-r} with n as the highest rank of an event can get among all components and r is the ranking of the particular event among all components. For instance, among the three events of death, MI, hospitalization in this order of clinical significance, the highest-ranking place one event can get is 3; for the death event, it is ranked as No.1, $r = 1$, thus a death event will get a score of $10^{3-1} = 100$. Similarly, the event of MI will yield a score of $10^{3-2} = 10$, while hospitalization will have a score of $10^{3-3} = 1$. Whenever a subject has an individual endpoint event, that subject gets the corresponding score per that event as discussed above. Otherwise, that subject gets a 0 score. The final composite score for each subject is the sum of individual event scores from all components. For example, if one subject has reported a death and a MI in the study, that subject will get a final composite score of 100+10 = 110. The following codes demonstrate how to derive this composite score:

```
data compscore_&k;
  set simulist_&k;
  %do i=1 %to %sysfunc(countw(&event));
  %let component = %scan(&event, &i);
     compscore_&i = &component. * (10 ** (%sysfunc(countw(&event)) - &i.));
  %end;

  compscore = sum (of compscore_:);
  keep subjid arm simn &event compscore;
run;
```

With the intermediate composite score derived as above, a comparison for each pair of subjects to determine a winner or a loser or a tie can be readily available as 1) if Subject A's composite score is greater than the Subject B's composite score in an given pair, then that Subject A is a loser and gets a score of -1; 2) if Subject A's composite score is equal to the Subject B's composite score in an given pair, then that Subject A is tied and gets a score of 0; 3) if Subject A's composite score is less than the Subject B's composite score in an given pair, then that Subject A is a winner and gets a score of 1. Through a *N x N* matrix, the final win ratio scores can be obtained in the following codes:

```
proc sort data = compscore_&k;
    by SUBJID;
run;

proc transpose data = compscore_&k out = compscore_wide_&k prefix = SUBJID;
    id SUBJID;
    var compscore;
run;

proc sql;
    create table compscore_matrix_&k as
    select
    *
    from compscore_&k, compscore_wide_&k
    ;
quit;

proc sql noprint; select COMPRESS("SCORE_"||SUBJID) into: SUBJIDlist
separated by ' ' from compscore_&k; quit;

data winscore_&k;
  set compscore_matrix_&k;
  array comp(%sysevalf(2 * &subn)) SUBJID_: ;
  array score(%sysevalf(2 * &subn)) &SUBJIDlist;

  do x = 1 to %sysevalf(2 * &subn);
    if COMPSCORE > comp(x) then score(x) = -1;
    if COMPSCORE = comp(x) then score(x) = 0;
    if COMPSCORE < comp(x) then score(x) = 1;
    if _n_ = x then score(x) = .;
  end;

  Ui = sum (of SCORE_:);
  if arm = "treatment" then Di = 1;
  if arm = "control" then Di = 0;
  UiDi = Ui * Di;
  Ui2 = Ui * Ui;
```

```
  keep subjid arm &event score_: Ui Di UiDi Ui2;
run;
```

Up till this point, the win-lose score U_i in formula ② is derived. With the score U_i ready, the U_iD_i required for further imputation in formula ③ and U_i^2 needed in formula ④ can also be imputed for the calculation of the win ratio statistics per formula ① above. The algorithms in the formulas ① ③ ④ can be translated into the SAS codes below where *T* and *V* respectively corresponds to the T score and variance while *FS_test* is the final F-S score (Finkelstein & Schoenfeld, 1999).

```
proc sql noprint;
    create table winstatistics_&k as
    select
    &k as simn
    , sum(UiDi) as T
    ,((&subn*&subn)/(%sysevalf(2 * &subn)*(%sysevalf(2 * &subn)-1))) *
(sum(UI2)) as V
    ,calculated T / sqrt(calculated V) as FS_test
    from winscore_&k
    ;
quit;

%end;
```

STEP 3. CALCULATE THE POWER

Before introducing the codes for power calculations, it is necessary to remind again that up to this time point, in real world practice, all simulation lists need to be verified on 1) if the frequency of composite events in each simulated listing reflects the pre-specified event proportions, and 2) if correlations among components within each simulated listing are reasonably low. The first verification can be analyzed through a simple ***proc freq*** statement, while the second can be done by a quick ***proc corr*** statement. This paper will not elaborate on these two relatively easy tasks to save space here.

As mentioned above, when the F-S score per each data replicate is calculated, the power of the study can be imputed by the following steps: 1) compare the F-S score to the standard normal distribution to determine the p value in each data replicate; 2) with α set to 0.05 in usual study designs, determine each simulation as whether its p value is on or below the threshold of 0.05; 3) derive the proportion of data replicates with p-values that are equal or below 0.05 based off all data replicates. The first two steps can be completed in one ***cdf*** function below:

```
data winstatistics_all;
  set winstatistics_:;
  p = 2 *(1- cdf('NORMAL',FS_test));
run;
```

Finally, by calculating the proportion of data replicates with $p \leq 0.05$ from all simulated data replicates, the power of the study can be obtained as:

```
proc sql;
   create table win_power as
   select
   sum(case when p le 0.05 then 1 else 0 end) as p05_sum
   , count (simn) as total
   , (calculated p05_sum / calculated total) as power
   from winstatistics_all
  ;
quit;
```

CONCLUSION

The win ratio approach in composite endpoint analysis is gaining steady momentum in cardiovascular trials. While more of those attention has been paid toward the final statistical analysis and hypothesis testing stage after study data are collected, this paper is among the first few to implement the win ratio approach at the early study design stage. By proposing this easy-to-use macro, the authors of this paper hope to make an once complex and challenging task now easy for biostatisticians who want to explore the win ratio approach in study design or potentially even in adapting to study analysis and hypothesis testing.

Of course, estimating the power of a study per simulation is usually not the only goal in study designing. There are many more questions to be answered and discussed. In the meantime, there are still limitations in this macro. For instance, this macro addresses only binary outcome events as they are most common composite endpoints in cardiovascular trials. However, in more complex scenarios and applications, multinominal events and/or continuous events could also be integrated into composite endpoint designs. Those complex scenarios are not in the capacity of this SAS macro at this point yet. Second, this macro does not take time-to-event into account when both subjects in a given pair have the same event for simplification purpose as this macro was originally designed to facilitate exploratory purpose for study designing. Thus, future research and discussions are necessary to extend and improve this macro to accommodate for more complex study designs.

REFERENCES

Abdalla, S., Montez-Rath, M. E., Parfrey, P. S., & Chertow, G. M. (2016). The win ratio approach to analyzing composite outcomes: An application to the evolve trial. *Contemporary Clinical Trials*, *48*, 119–124. https://doi.org/10.1016/j.cct.2016.04.001

Andersen, P. K., & Gill, R. D. (1982). Cox's regression model for counting processes: A large sample study. *The Annals of Statistics*, *10*(4). https://doi.org/10.1214/aos/1176345976

Bakal, J. A., Roe, M. T., Ohman, E. M., Goodman, S. G., Fox, K. A., Zheng, Y., Westerhout, C. M., Hochman, J. S., Lokhnygina, Y., Brown, E. B., & Armstrong, P. W. (2015). Applying novel methods to assess clinical outcomes: Insights from the trilogy ACS trial. *European Heart Journal*, *36*(6), 385–392. https://doi.org/10.1093/eurheartj/ehu262

Bebu, I., & Lachin, J. M. (2018). Properties of composite time to first event versus joint marginal analyses of multiple outcomes. *Statistics in Medicine*, *37*(27), 3918–3930. https://doi.org/10.1002/sim.7849

Cox, D. R. (1972). Regression models and life-tables. *Journal of the Royal Statistical Society: Series B (Methodological)*, 187–202. https://doi.org/10.1111/j.2517-6161.1972.tb00899.x

Dong, G., Qiu, J., Wang, D., & Vandemeulebroecke, M. (2018). The stratified win ratio. *Journal of Biopharmaceutical Statistics*, *28*(4), 778–796. https://doi.org/10.1080/10543406.2017.1397007

Ferreira, J. P., Kraus, B. J., Zwiener, I., Lauer, S., Zinman, B., Fitchett, D. H., Koitka-Weber, A., George, J. T., Ofstad, A. P., Wanner, C., & Zannad, F. (2021). Cardio/Kidney Composite End Points: A post hoc analysis of The empa-reg outcome trial. *Journal of the American Heart Association*, *10*(7). https://doi.org/10.1161/jaha.120.020053

Finkelstein, D. M., & Schoenfeld, D. A. (1999). Combining mortality and longitudinal measures in clinical trials. *Statistics in Medicine*, *18*(11), 1341–1354. https://doi.org/10.1002/(sici)1097-0258(19990615)18:11<1341::aid-sim129>3.0.co;2-7

Kaplan, E. L., & Meier, P. (1958). Nonparametric estimation from incomplete observations. *Journal of the American Statistical Association*, *53*(282), 457–481. https://doi.org/10.1080/01621459.1958.10501452

Mantel N. (1966). Evaluation of survival data and two new rank order statistics arising in its consideration. *Cancer chemotherapy reports*, *50*(3), 163–170.

Peng, L. (2020). The use of the win odds in the design of non-inferiority clinical trials. *Journal of Biopharmaceutical Statistics*, *30*(5), 941–946. https://doi.org/10.1080/10543406.2020.1757690

Pocock, S. J., Ariti, C. A., Collier, T. J., & Wang, D. (2012). The win ratio: A new approach to the analysis of composite endpoints in clinical trials based on clinical priorities. *European Heart Journal*, *33*(2), 176–182. https://doi.org/10.1093/eurheartj/ehr352

Rogers, J. K., Pocock, S. J., McMurray, J. J. V., Granger, C. B., Michelson, E. L., Östergren, J., Pfeffer, M. A., Solomon, S. D., Swedberg, K., & Yusuf, S. (2013). Analysing recurrent hospitalizations in heart failure: A review of statistical methodology, with application to charm-preserved. *European Journal of Heart Failure*, *16*(1), 33–40. https://doi.org/10.1002/ejhf.29

Wei, L. J., & Glidden, D. V. (1997). An overview of statistical methods for multiple failure time data in clinical trials. *Statistics in Medicine*, *16*(8), 833–839. https://doi.org/10.1002/(sici)1097-0258(19970430)16:8<833::aid-sim538>3.0.co;2-2

Wei, L. J., Lin, D. Y., & Weissfeld, L. (1989). Regression analysis of multivariate incomplete failure time data by modeling marginal distributions. *Journal of the American Statistical Association*, *84*(408), 1065–1073. https://doi.org/10.1080/01621459.1989.10478873

Yosef, M., Khalatbari, S., & Hummel, S. (2019, October 20–22). *Extending the Finkelstein-Schoenfeld test and Win Ratio to three composite Outcomes Using SAS* [Paper presentation]. SESUG 2019, Williamsburg, VA, United States.

APPENDIX

```
%macro winPower (event =, ratio1 =, ratio2 =, subn =, simnum = );
%local i j k component proportion;
%do k = 1 %to &simnum;
      %do m = 1 %to 2;
            %do i = 1 %to %sysfunc(countw(&event));
            %let component = %scan(&event, &i);
            %let proportion = %scan(&&ratio&m, &i, '|');

/**************STEP 1. PRODUCE SUBJECT-LEVEL LISTING*************/

                  data &component._temp;
                   do j = 1 to &subn;
                    if j le ROUND(&subn. * &proportion.) then &component = 1;
                    else &component = 0;
                    x1 = rand('normal',.5,.1);
                    output;
                   end;
               run;

                  proc sort data = &component._temp;
                        by x1;
                  run;

                  data &component;
                        length SUBJID arm $20.;
                        format simn best.;
                        set &component._temp;
                        by x1;
                        retain SUBJID_;
                        if _n_ = 1 then SUBJID_ = &m. * &subn. + 1;
                        else SUBJID_ = SUBJID_ +1;

                        SUBJID = strip(put(SUBJID_, best.));
                        simn = &k;
                        if &m = 1 then arm = "treatment";
                        if &m = 2 then arm = "control";

                        keep SUBJID arm simn &component;
                  run;

                  proc sort data = &component;
                        by SUBJID;
                  run;
            %end;
            data arm_&k._&m;
                  merge &event;
                  by SUBJID;
            run;

      %end;
      data simulist_&k;
            set arm_&k.:;
      run;
```

```
/**************STEP 2. DERIVE WIN RATIO STATISTICS**********************/

data compscore_&k;
      set simulist_&k;
      %do i = 1 %to %sysfunc(countw(&event));
            %let component = %scan(&event, &i);
            compscore_&i =  &component. * (10 ** (%sysfunc(countw(&event)) -
&i.));
      %end;

      compscore = sum (of compscore_:);
      keep subjid arm simn &event compscore;
run;

proc sort data = compscore_&k;
      by SUBJID;
run;

proc transpose data=compscore_&k out=compscore_wide_&k prefix=SUBJID_;
    id SUBJID;
    var compscore;
run;

proc sql;
      create table compscore_matrix_&k as
      select
      *
      from compscore_&k, compscore_wide_&k
      ;
quit;

proc sql noprint; select COMPRESS("SCORE_"||SUBJID) into: SUBJIDlist
separated by ' ' from compscore_&k; quit;

data winscore_&k;
      set compscore_matrix_&k;
      array comp(%sysevalf(2 * &subn)) SUBJID_: ;
      array score(%sysevalf(2 * &subn)) &SUBJIDlist;

      do x = 1 to %sysevalf(2 * &subn);
            if COMPSCORE > comp(x) then score(x) = -1;
            if COMPSCORE = comp(x) then score(x) = 0;
            if COMPSCORE < comp(x) then score(x) = 1;
            if _n_ = x then score(x) = .;
      end;

      Ui = sum (of SCORE_:);
      if arm = "treatment" then Di = 1;
      if arm = "control" then Di = 0;
      UiDi = Ui* Di;
      Ui2 = Ui * Ui;

      keep subjid arm &event score_: Ui Di UiDi Ui2;
run;

proc sql noprint;
      create table winstatistics_&k as
```

```
      select
      &k as simn
      , sum(UiDi) as T
      ,((&subn*&subn)/(%sysevalf(2 * &subn)*(%sysevalf(2 * &subn)-1))) *
(sum(UI2)) as V
      ,calculated T / sqrt(calculated V) as FS_test
      from winscore_&k
      ;
quit;

%end;

/****CODES FREQUENCY AND CORRELATION CHECKING OMMITTED HERE***/


/*******STEP 3. CALCULATE THE POWER ******************************/
data winstatistics_all;
      set winstatistics_:;
      p = 2 *(1- cdf('NORMAL',FS_test));
run;

proc sql;
      create table win_power as
      select
      sum(case when p le 0.05 then 1 else 0 end) as p05_sum
      , count (simn) as total
      , (calculated p05_sum / calculated total) as power
      from winstatistics_all
      ;
quit;

/****CODES FOR OUTPUTTING DATA SETS TO LOCAL FOLDERS OMMITTED HERE******/

%mend;
```

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Name: Lili Li
Enterprise: Inari Medical
Address: 6001 Oak Canyon, Irvine, CA 92618
Email: lili.li@inarimedical.com

Name: Chih-Chieh Chang
Enterprise: Inari Medical
Address: 6001 Oak Canyon, Irvine, CA 92618
Email: roger.chang@inarimedical.com

Name: Benjamin Fine
Enterprise: Inari Medical
Address: 6001 Oak Canyon, Irvine, CA 92618
Email: benjamin.fine@inarimedical.com

Name: Yu-Chen Su
Enterprise: Inari Medical
Address: 6001 Oak Canyon, Irvine, CA 92618
Email: ray.su@inarimedical.com